



In deze toepassing voor module 2 ga je programma's schrijven om enkele wiskundige formules uit te werken.

#### Doelen:

- De TI Basic opdrachten die geleerd zijn in module 2 gebruiken om een programma te schrijven dat een formule uitwerkt.

### De stelling van Pythagoras

In een rechthoekige driehoek met rechthoekszijden A en B (in het Engels de legs) en schuine zijde (hypotenusa) C geldt:

$$A^2 + B^2 = C^2$$

Schrijf een programma dat de gebruiker vraagt om de lengtes van de rechthoekszijden in te voeren en dan de lengte van de schuine zijde berekent en alle drie de waarden netjes weergeeft op het scherm.

Gebruik het Engelse scherm als voorbeeld.

```
NORMAL FLOAT AUTO REAL RADIAN MP
PROGRAM:PYTHAG
:ClrHome
:Disp "THIS PROGRAM COMPUT
ES"
:Disp "THE HYPOTENUSE"
:Disp "ENTER THE LEGS..."
:Prompt A,B
:
:
:
```

*Opmerking: je moet eerst de formule hierboven herschrijven in de vorm C = ...*

### De formule van Heron

**De formule van Heron** bepaalt de oppervlakte van elke willekeurige driehoek uitsluitend op basis van de lengtes van de drie zijden A, B en C van de driehoek. De formule wordt meestal in twee delen beschreven.

$S = (A + B + C) / 2$  is de halve omtrek van de driehoek

$A = \sqrt{S * (S - A)(S - B)(S - C)}$  is de oppervlakte van de driehoek

```
NORMAL FLOAT AUTO REAL RADIAN MP
PROGRAM:HERON
:ClrHome
:Disp "THIS PROGRAM COMPUT
ES"
:Disp "HERON'S FORMULA"
:Disp "ENTER THE SIDES..."
:
:Prompt A,B,C
:
:
:
```

Schrijf een programma dat de gebruiker vraagt om de lengtes van de drie zijden van een driehoek in te voeren en vervolgens de oppervlakte berekent en de zijden en de oppervlakte weergeeft op een mooi scherm.

Gebruik het Engelse scherm als voorbeeld.

*Opmerking: de gebruiker kan drie waarden invoeren die niet de lengten van de zijden van een driehoek kunnen zijn. Wat gebeurt er als de gebruiker ongeldige waarden invoert?*

**Docent Tip:** De formule van Heron werkt niet (geeft een niet-reëel antwoord) als de drie waarden onmogelijk de lengten van de zijden van één driehoek kunnen zijn (de driehoeksongelijkheid). Hier kan rekening mee worden gehouden door de complexe modus te veranderen in a + bi..

# 10 minuten programmeren

## TI-84 PLUS SERIE

### De abc-formule (quadratic formula)

Als een kwadratische vergelijking de vorm  $Ax^2 + Bx + C = 0$  heeft, dan kunnen de oplossingen van de vergelijking bepaald worden door .....

Eerst de discriminant te bepalen:

$$D = B^2 - 4AC$$

En dan zijn de twee oplossingen:

$$R1 = (-B + \sqrt{D}) / (2A)$$

$$R2 = (-B - \sqrt{D}) / (2A)$$

Schrijf een programma dat de gebruiker vraagt om de drie coëfficiënten A, B en C van de kwadratische vergelijking in te voeren en dat de coëfficiënten en de twee oplossingen van de vergelijking netjes weergeeft.

Gebruik het Engelse scherm als voorbeeld.

*Opmerking: Je kunt R1 en R2 niet gebruiken als variabelen! Gebruik iets anders. Wat kan er fout gaan bij dit programma?*

## MODULE 2: TOEPASSING

### DOCENTENHANDLEIDING

```
NORMAL FLOAT AUTO REAL RADIAN MP
PROGRAM:QUAD
:ClrHome
:Disp "THIS PROGRAM COMPUT
ES THE"
:Disp "QUADRATIC FORMULA"
:Disp "ENTER THE COEFFICIE
NTS..."
:Prompt A,B,C
:
:
```

**Docent Tip:** Hieronder staan de programmaregels voor elke opdracht. De belangrijkste stappen zijn de berekeningen met de formule. De output-posities zijn geschikt voor elke TI-84, maar de TI-84 C/CE kan andere waarden gebruiken afhankelijk van het gewenste schermbeeld.

#### De stelling van Pythagoras

##### Antwoord:

```
prgmPYTHAG
ClrHome
Disp "THIS PROGRAM COMPUTES"
Disp "THE HYPOTENUSE"
Disp "ENTER THE LEGS..."
Prompt A,B
√(A²+B²)→C

ClrHome
Output(3,5,"A= ")
Output(3,8,"A)
Output(4,5,"B= ")
Output(4,8,B)
Output(6,5," HYPOTENUSE = ")
Output(6,16,C)
Pause
ClrHome
```

#### De formule van Heron

##### Antwoord:

```
prgmHERON
ClrHome
Disp "THIS PROGRAM COMPUTES"
Disp "HERON'S FORMULA"
Disp "ENTER THE SIDES..."
Prompt A,B,C
(A+B+C)/2→S
√(S(S-A)(S-B)(S-C))→D

ClrHome
Output(3,5,"A= ")
Output(3,8,A)
Output(4,5,"B= ")
Output(4,8,B)
Output(5,5,"C= ")
Output(5,8,C)
Output(7,5,"AREA= ")
Output(7,11,D)
Pause
ClrHome
```

#### De abc-formule

##### Antwoord:

```
prgmQUAD
ClrHome
Disp "THIS PROGRAM COMPUTES
THE"
Disp "QUADRATIC FORMULA"
Disp "ENTER THE COEFFICIENTS..."
Prompt A,B,C
B²-4AC→D
(-B+√(D))/(2A)→R
(-B-√(D))/(2A)→S

ClrHome
Output(3,5,"A= ")
Output(3,8,A)
Output(4,5,"B= ")
Output(4,8,B)
Output(5,5,"C= ")
Output(5,8,C)
Output(7,5,"ROOT1= ")
Output(7,12,R)
Output(8,5,"ROOT2= ")
Output(8,12,S)
Pause
```



# 10 minuten programmeren

TI-84 PLUS SERIE

MODULE 2: TOEPASSING

DOCENTENHANDLEIDING

ClrHome